Problem:
Score : 400 points

Problem Statement

We have an undirected graph G with N vertices numbered 1 to N and N edges as follows:

For each i=1,2,...,N-1, there is an edge between Vertex i and Vertex i+1.

There is an edge between Vertex X and Vertex Y.

For each k=1,2,...,N-1, solve the problem below:

Find the number of pairs of integers (i,j) (1 \leq i < j \leq N) such that the shortest distance between Vertex i and Vertex j in G is k.

Constraints

3 \leq N \leq 2 \times 10^3

1 \leq X,Y \leq N

X+1 < Y

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N X Y

Output

For each k=1, 2, ..., N-1 in this order, print a line containing the answer to the problem.

Sample Input 1

5 2 4

Sample Output 1

5
4
1
0

The graph in this input is as follows:

There are five pairs (i,j) (1 \leq i < j \leq N) such that the shortest distance between Vertex i and Vertex j is 1: (1,2)\,,(2,3)\,,(2,4)\,,(3,4)\,,(4,5).

There are four pairs (i,j) (1 \leq i < j \leq N) such that the shortest distance between Vertex i and Vertex j is 2: (1,3)\,,(1,4)\,,(2,5)\,,(3,5).

There is one pair (i,j) (1 \leq i < j \leq N) such that the shortest distance between Vertex i and Vertex j is 3: (1,5).

There are no pairs (i,j) (1 \leq i < j \leq N) such that the shortest distance between Vertex i and Vertex j is 4.

Sample Input 2

3 1 3

Sample Output 2

3
0

The graph in this input is as follows:

Sample Input 3

7 3 7

Sample Output 3

7
8
4
2
0
0

Sample Input 4

10 4 8

Sample Output 4

10
12
10
8
4
1
0
0
0

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import qualified Data.Vector.Unboxed.Mutable as VUM
import qualified Data.Vector.Unboxed as VU
import qualified Data.Set as ST
import qualified Data.Map.Strict as MPS
import Data.List
import Data.Tuple
import Control.Monad

type Graph = MPS.Map Int [(Int, Int)]
type Distance = VUM.IOVector Int
type Queue = ST.Set (Int, Int)

makeGraph :: [(Int, [(Int, Int)])] -> Graph
makeGraph xs = MPS.fromListWith (++) xs

dijkstra :: Int -> Int -> Graph -> IO Distance
dijkstra vertex from graph = do
  distance <- VUM.replicate (vertex+1) (10^9) :: IO Distance
  VUM.write distance from 0 
  trav graph distance (ST.singleton (0, from))
  where
    trav :: Graph -> Distance -> Queue -> IO Distance
    trav graph distance queue
      | ST.null queue = return distance
      | otherwise = do
          tmp <- VUM.read distance from
          if acc <= tmp
            then update >>= trav graph distance
            else trav graph distance queue'
        where
          ((acc, from), queue') = ST.deleteFindMin queue

          update :: IO Queue
          update = 
            case MPS.lookup from graph of
              Just xs -> update' xs queue'
              Nothing -> return queue'
            where
              update' :: [(Int, Int)] -> Queue -> IO Queue
              update' [] q = return q
              update' ((to, cost):xs) q = do
                tmp <- VUM.read distance to
                let acc' = acc + cost
                case compare acc' tmp of
                   LT -> do
                      VUM.write distance to acc'
                      update' xs (ST.insert (acc', to) q)
                   _ ->
                      update' xs q
        
main = do
  [n,a,b] <- map read . words <$> getLine :: IO [Int]
  let vs = let xs = zip [1..n] [2..n] in (map swap xs) ++ xs :: [(Int, Int)]
  let vs' = [(a,[(b,1)]), (b,[(a,1)])] ++ map (\(f,t) -> (f,[(t,1)])) vs :: [(Int, [(Int, Int)])]
  let graph = makeGraph vs'

  table <- VUM.replicate n 0 :: IO (VUM.IOVector Int)

  forM_ [1..n] $ \i -> do
    dist <- dijkstra n i graph
    dist' <- VU.tail . VU.zip (VU.fromList [0..n]) <$> VU.unsafeFreeze dist
    VU.forM_ dist' $ \(j,d) ->
      when (d /= 10^9 && j > i) $  
        VUM.modify table (+1) (d-1)

  table' <- VU.init <$> VU.unsafeFreeze table
  VU.mapM_ print table'